
/**************************************************************************** 
                 ApacheII  Macro 
     Predicted Death Rates based on APACHE II 
     
Modified by Kendiss Olafson October 2005 to convert into the new codes 
 
Last modified by Julie: April 17, 2007 with additional diagnostic codes 
 
****************************************************************************/; 
 
%macro apacheII; 
 
data apa; set mod.apache(keep=study es); 
run; 
 
proc sort data=apa; by study; 
run; 
 
* ASSIGN NONOPERATIVE AND POSTOPERATIVE PATIENTS; 
Data apadx; set mod.reg(keep=study from type adm1-adm6 apache cm1-cm6); 
run; 
 
proc sort data=apadx; by study; 
run; 
 
data apadx; merge apadx apa; 
  by study; 
  
  prevLOC = substr(FROM,2,1); 
  IF prevLOC = 'P' OR (prevloc = 'R' AND type = 'S') then postop = 1;  
  else postop = 0; 
  IF Postop=1 AND ES = 'N' then EmergSurg = 1;  
  else Emergsurg = 0; 
 
  ** ASSIGN PRINCIPAL DIAGNOSTIC CATEGORIES; 
 
  index1=0;index2=0;index3=0;index4=0;index5=0;index6=0; 
 
        ARRAY ndex{6} index1 --index6; 
        ARRAY prindx{6} adm1 -- adm6; 
 
*       DEFINE ASPIR/POISON; 
        aspir=0;poison=0; 
 
  Do g = 1 to 6; 
                IF 450 LE prindx{g} LE 499.9 
                                THEN poison=1; 
              IF prindx{g} IN (36.59,37.59, 38.59, 39.59,2) 
                                THEN aspir=1; 
        End;        
        asppoi= aspir + poison*2; 
 
*  DEFINE Cancer and Cabg; 
  ca = 0; cabg = 0; 
  Do h = 1 to 6; 
   if  700 le prindx{h} le 749.9 
       THEN ca = 1; 



   If prindx{h} in (750, 750.01, 750.02)  
       THEN cabg = 1; 
  END; 
*   
 
/*****          NONOPERATIVE PATIENTS           *****/; 
 
iF postop=0 THEN DO; 
 DO i = 1 to 6; 
  *resp failure from ca; 
         If prindx{i} in (23.06, 28.02,30.07,30.15) 
                THEN ndex{i}=29; 
  *dissecting thoracic/abdominal aneurysm; 
         ELSE iF 127 <= prindx{i} AND prindx{i} <=127.9  
                THEN ndex{i}=28; 
  *SDH/epidural/intracerebral/SAH/brainstem bleed; 
         ELSE iF prindx{i}in (500, 500.01, 501, 502, 502.01, 502.03, 502.04, 502.05, 
502.9, 503, 
     503.01, 503.9, 504, 504.01, 504.02, 504.9)  
                THEN ndex{i}=27; 
  *hypovolemic/hemorrhagic shock; 
         ELSE iF prindx{i} iN (136,136.01,136.02,136.03,136.04,136.05,136.06, 
136.07,136.9) 
                THEN ndex{i}=25; 
  *cardiac arrest; 
         ELSE iF prindx{i} iN (100,100.01,100.02, 130.01, 130.02, 131.04) 
                THEN ndex{i}=23; 
  *gi bleeding; 
         ELSE iF 401 <= prindx{i} AND prindx{i} <= 402.9   
    THEN ndex{i}=22; 
  *sepsis; 
         ELSE iF (44 le prindx{i} le 46.91) OR (prindx{i} in (89, 90)) 
    THEN ndex{i}=21; 
  *respiratory infection; 
         ELSE iF (36 le prindx{i} le 39.91) OR (prindx{i} in 
(36.59,37.59,38.59,39.59)AND poison=0) 
                THEN ndex{i}=20; 
  *pulmonary embolism; 
         ELSE iF 13 le prindx{i} le 13.9 
                THEN ndex{i}=19; 
  *aspiration/poisoning; 
         ELSE iF asppoi=3 
                THEN ndex{i}=18; 
  *respiratory arrest; 
         ELSE iF prindx{i}=100.03 
                THEN ndex{i}=17; 
  *coronary artery disease=Mi,ischemic cardio-m,unstable angina,acute coronary 
                insufficiency,post-infarct angina; 
         ELSE iF (102 le prindx{i}le 102.1) OR (prindx{i}iN (104.01, 151, 152, 155)) 
                THEN ndex{i}=16; 
  *non-cardiogenic pulmonary edema; 
         ELSE iF prindx{i}=2 
                THEN ndex{i}=15; 
  *cardiogenic shock; 
         ELSE iF prindx{i} iN (101,101.01,101.02,101.03,101.04,101.05,101.06,101.08, 
101.9) 
                THEN ndex{i}=14; 



  *resp failure from COPD; 
         ELSE iF prindx{i}iN (3, 8) 
                THEN ndex{i}=13; 
  *congestive heart failure; 
         ELSE iF prindx{i} iN (103,103.01,103.02,103.03,103.04,103.05,103.06,103.07, 
103.9) 
                THEN ndex{i}=12; 
  *head trauma=concussion,contusion,sub-dural,skull#,SAH,brainstem 
bleed,epidural; 
         ELSE iF prindx{i} iN (505.03, 508.04) OR (600 le prindx{i} le 600.1)  
     OR (602 le prindx{i} le 602.1)OR (647 le prindx{i} le 
647.1)  
     OR (605 le prindx{i} le 605.1)OR (648 le prindx{i} le 
648.1)  
     OR (654 le prindx{i} le 654.1)                                     
                THEN ndex{i}=11;                                                                
 *seizure disorder; 
         ELSE iF prindx{i} iN (509,509.02,509.03,509.04,509.05,509.07,508.08,509.09, 
                509.1,509.11,509.12,509.13,509.9) 
                THEN ndex{i}=10; 
 *multiple trauma; 
         ELSE iF 600  LE prindx{i} LE 699                                                      
                THEN ndex{i}=6;                                                                 
 *rhythm disturbance; 
         ELSE iF prindx{i} iN (130,130.03,130.04, 130.05,130.9,131,131.01,131.02, 
                131.03,131.9,131.97) 
                THEN ndex{i}=5; 
 *dka; 
         ELSE iF prindx{i}=301 
                THEN ndex{i}=4; 
 *hypertension; 
         ELSE iF prindx{i}iN (199.19, 124, 124.9) 
                THEN ndex{i}=3; 
 *asthma; 
         ELSE iF prindx{i} iN (6,6.9) 
                THEN ndex{i}=2; 
 *drug overdose; 
         ELSE iF 450 LE prindx{i} LE 499.90 
                THEN ndex{i}=1; 
 *Gi NOS;                                                                                  
         ELSE iF 400 LE prindx{i} LE 449.9 
                THEN ndex{i}=26; 
 *cardiovascular NOS; 
         ELSE iF (40 LE prindx{i} LE 43.91) OR (104 LE prindx{i} LE 126) OR  
    (132 LE prindx{i} LE 134.9) OR (140 le prindx{i} le 141) 
    OR (766 le prindx{i} le 766.01)  
                THEN ndex{i}=24; 
 *neuro NOS; 
         ELSE iF (77 LE prindx{i} LE 80.91)OR (514 LE prindx{i} LE 599.04) 
                OR (599.06 LE prindx{i} LE 599.94)OR prindx{i} iN (25, 25.01, 505, 
505.01, 505.04,  
    505.06, 505.07. 505.08, 505.9, 506, 507, 507.9, 508, 508.01, 
508.03, 508.05, 508.9) 
                THEN ndex{i}=9; 
 *metabolic/renal NOS; 
         ELSE iF (51 le prindx{i} le 56.91) OR (302 le prindx{i} le 399.9) 
                THEN ndex{i}=8; 



 *respiratory NOS; 
        /* ELSE iF 1 LE prindx{i} LE 99.9 old*/ 
   ELSE iF (9 LE prindx{i} LE 11.9) OR (14 le prindx{i} le 23.04) OR 
     (23.07 LE prindx{i} LE 23.9) OR (28.03 le prindx{i} le 
30.06) OR 
     (30.08 LE prindx{i} LE 30.14) OR (30.16 le prindx{i} le 
35.59) OR 
     prindx{i} in (4, 4.01, 5, 7, 26, 26.03, 27, 28)  
                THEN ndex{i}=7; 
 
 END; 
END; 
 
/***********    POST-OPERATIVE PATIENTS         ********/ 
ELSE IF postop=1 THEN DO; 
 Do j = 1 to 6; 
  *cardiac arrest; 
         IF prindx{j} IN (100,100.01,100.02, 130.01, 130.02, 131.04) 
                THEN ndex{j}=30; 
  *sepsis; 
         ELSE IF (44 le prindx{j} le 46.91) or prindx{j}in (89,90)   
    THEN ndex{j}=31; 
  *GI perforation or obstruction; 
   Else if (69 le prindx{j} le 71.91) or  prindx{j} in (794.12, 418)  
    Then ndex{j} =32; 
  *Resp Insufficiency post-op; 
   Else if (24 le prindx{j} le 24.9) OR (38 le prindx{j} le 38.91) OR 
    prindx{j} in  (26.01, 100.03)   
    Then ndex{j} = 33; 
  *GI surgery for neoplasm; 
   Else if ca = 1 AND (prindx{j} in (784, 784.9, 785, 785.9, 786, 786.9, 
787, 787.9, 788, 788.9, 
   789, 789.9, 790, 790.9, 791, 791.9, 792, 792.9, 793, 793.9, 794.03, 
796, 796.9, 813, 813.9))  
   Then ndex{j} = 35; 
  *gi bleeding; 
         ELSE IF (401 le prindx{j} le 402.9) OR prindx{j} IN (794.15, 449.22) 
                THEN ndex{j}=38; 
  *hypovolemic/hemorrhagic shock; 
         ELSE IF prindx{j} IN (136,136.01,136.02,136.03,136.04,136.05,136.06, 
                136.07,136.9) 
                THEN ndex{j}=39; 
  *Laminectormy and other spinal surgery; 
   ELSE IF prindx{j} IN (830, 830.9, 831.02, 831.08) 
    THEN ndex{j} = 40; 
  *Craniotomy for ICH/SDH/SAH; 
   ELSE if (601 le prindx{j} le 601.1) OR (603 le prindx{j} le 604.1) OR  
    (653 le prindx{j} le 653.1) OR  
   prindx{j} IN (828.03, 828.05, 828.06, 828.08, 828.12, 502.02, 503.02) 
    THEN ndex{j} = 41; 
  *Thoracic surgery for neoplasm; 
   ELSE if ca = 1 AND ((769 le prindx{j}le 774) OR  prindx{j}IN (776, 
777.01))  
    THEN ndex{j} = 43; 
  *Head trauma; 
   ELSE if (prindx{j} ge 600 AND prindx{j} le 605.9) 
   OR (prindx{j} ge 648 AND prindx{j} le 648.9) 



   OR (prindx{j} ge 653 AND prindx{j} le 654.9) 
    THEN ndex{j} = 44; 
  *Renal transplant; 
   ELSE if prindx{j} in (805, 805.01, 805.02, 805.03) 
    then ndex{j} = 45; 
  *Craniotomy for neoplasm; 
   ELSE if ca = 1 AND prindx{j} in (828.02, 828.13, 828.9) 
    then ndex{j} = 47; 
  *Renal surgery for neoplasm; 
   ELSE if (ca = 1 and prindx{j} in (801.02, 801, 801.9)) OR prindx{j} = 
801.02 
    then ndex{j} = 48; 
  *Heart Valve Surgery; 
   ELSE if (751 le prindx{j} le 752.9) 
    OR prindx{j}in (754, 754.01, 754.02, 755, 758, 758,01, 758.02, 
768) 
    THEN ndex{j} = 49; 
  *Peripheral vascular surgery; 
   ELSE if (753 le prindx{j} le 753.9) OR (760 le prindx{j} le 760.9) OR 
    prindx{j} IN (156, 156.01, 756, 757, 763.04, 763.07, 763.09, 
763.22) 
    THEN ndex{j} = 50; 
     
  *Multiple Trauma; 
   Else if 600 le prindx{j} le 661.9  
    THEN ndex{j} = 52; 
  *metabolic/renal NOS; 
         ELSE IF (301 LE prindx{j} LE 399.9) OR (54 le prindx{j} le 56.91) 
                THEN ndex{j}=34; 
  *respiratory NOS; 
         ELSE IF (2 LE prindx{j} LE 23.9) OR (27 LE prindx{j} LE 37.97) OR  
    (prindx{j} IN (26, 26.01, 26.03, 781, 781.01, 781.02)) 
                THEN ndex{j}=36; 
  *GI NOS; 
         ELSE IF (57 LE prindx{j} LE 68.97) OR (72 LE prindx{j} LE 76.97) 
    OR (400 LE prindx{j} LE 449.97)  
                THEN ndex{j}=37; 
  *cardiovascular NOS; 
         ELSE IF (101 LE prindx{j} LE 134.9) OR prindx{j} IN (140, 141, 151, 152, 155, 
898, 24.12)  
                THEN ndex{j}=42; 
  *neuro NOS; 
         ELSE IF (500 LE prindx{j} LE 599.9) 
                OR prindx{j} IN (209,852, 25, 25.01) 
                THEN ndex{j}=46; 
  
 END; 
END; 
  
drop g h i j; 
 
index = 0; 
IF index1 > 0 then index = index1; 
 else if index2 > 0 then index = index2; 
 else if index3 > 0 then index = index3; 
 else if index4 > 0 then index = index4; 
 else if index5 > 0 then index = index5; 



 else if index6 > 0 then index = index6; 
*Admission due to chronic CV - NEED TO STILL DEFINE; 
IF postop =1 AND index = 0 THEN DO; 
 ARRAY CM{6}  CM1 CM2 CM3 CM4 CM5 CM6; 
 Do k = 1 to 6; 
 IF CM{k} IN (151, 152, 158, 102, 103, 104, 130, 131, 128, 766) THEN index = 51; 
 END; 
END; 
drop k; 
 
run; 
 
Data apadx; 
 set apadx; 
 
 array xs {52} w1-w52; 
 do i=1 to 52; 
     if index=i then xs{i}=1; else xs{i}=0; 
    end; 
  
 APA= -3.517 + (apache*0.146) + (0.603 * Emergsurg)  
       - (w1*3.353)  - (w2*2.108)  - (w3*1.798)  - (w4*1.507)  - (w5*1.368) 
    - (w6*1.228)  - (w7*0.890)  - (w8*0.885)  - (w9*0.759)  - (w10*0.584) 
       - (w11*0.517) - (w12*0.424) - (w13*0.367) - (w14*0.259) - (w15*0.251)    
       - (w16*0.191) - (w17*0.168) - (w18*0.142) - (w19*0.128) - (w20*0.0) 
    + (w21*0.113) + (w22*0.334) + (w23*0.393) + (w24*0.470) + (w25*0.493) 
       + (w26*0.501) + (w27*0.723) + (w28*0.731) + (w29*0.891) + (w30*0.393) 
  + (w31*0.113) + (w32*0.060) - (w33*0.140) - (w34*0.196) - (w35*0.248) 
  - (w36*0.610) - (w37*0.613) - (w38*0.248) - (w39*0.682) - (w40*0.699) 
  - (w41*1.245) - (w42*0.797) - (w43*0.802) - (w44*0.955) - (w45*1.042) 
  - (w46*1.150) - (w47*1.245) - (w48*1.204) - (w49*1.261) - (w50*1.315) 
  - (w51*1.376) - (w52*1.684); 
 
 expAPA=exp(APA); 
 PredMort= expAPA / ( 1+ expAPA); 
 
drop i; 
 
 keep study index APA expAPA PredMort; 
run; 
 
proc sort data=apadx out=mod.apadx; 
  by study; 
run; 
 
 
proc freq data=apadx order=freq; 
  table index /list; 
run; 
 
 
 
%mend apacheII; 
 
%apacheII; 
 


